



Enhancing the Plot

SP4R04d03.sas


1. Create a data table called **Sales** with 12 observations corresponding to a 12-month period. In addition, create a variable called **Revenue**. This variable is randomly generated from a normal distribution with a mean of 10,000 and a standard deviation of 1,000. Create a second variable called **Revenue_2**. This variable is also randomly generated from a normal distribution with a mean of 13,000 and a standard deviation of 500.

```
data sp4r.sales;
  call streaminit(123);
  do month=1 to 12;
    revenue = rand('Normal',10000,1000);
    revenue_2 = rand('Normal',13000,500);
    output;
  end;
run;
```

2. Use PROC SGPLOT to create a series plot with both variables.
 - a. Provide the title 'Monthly Sales of Company A and B for 2015', the X-axis label 'Month', the Y-axis label 'Revenue for 2015', a vertical reference line at the 6.5 month mark, and a horizontal reference line at 11,000.
 - b. Put your name in the bottom right corner with the INSET statement.
 - c. Provide tick marks from 1 to 12 by 1 for the X-axis.
 - d. Create the legend labels 'Company A' and 'Company B', with a blue line and a red line to distinguish them. Give each line a dashed pattern.

```
proc sgplot data=sp4r.sales;
  series x=month y=revenue / legendlabel='Company A'
    lineattrs=(color=blue pattern=dash);
  series x=month y=revenue_2 / legendlabel='Company B'
    lineattrs=(color=red pattern=dash);

  title 'Monthly Sales of Company A and B for 2015';
  xaxis label="Month" values=(1 to 12 by 1);
  yaxis label="Revenue for 2015";
  inset "Jordan Bakerman" / position=bottomright;
  refline 6.5 / transparency= 0.5 axis=x;
  refline 11000 / transparency= 0.5;
run;
```



3. Use the **Sales** data table to further enhance the plot from the previous demonstration.
 - a. Use the SCATTER statement to add the observations to the plot.
 - b. Use the SYMBOL=circlefilled option for the points.
 - c. Add the KEYLEGEND statement to avoid multiple legends for both the SCATTER and SERIES statements.
 - d. Change the Y-axis limits to 8000 and 14000.

```
proc sgplot data=sp4r.sales;
  series x=month y=revenue / legendlabel='Company A' name='Company A'
    lineattrs=(color=blue pattern=dash);
  scatter x=month y=revenue / markerattrs=(color=blue
    symbol=circlefilled);
  series x=month y=revenue_2 / legendlabel='Company B'
    name='Company B' lineattrs=(color=red pattern=dash);
  scatter x=month y=revenue_2 / markerattrs=(color=red
    symbol=circlefilled);

  title 'Monthly Sales of Company A and B for 2015';
  xaxis label="Month" values=(1 to 12 by 1);
  yaxis label="Revenue for 2015" min=8000 max=14000;
  inset "Jordan Bakerman" / position=bottomright;
  refline 11000 / transparency= 0.5;
  refline 6.5 / transparency= 0.5 axis=x;
  keylegend 'Company A' 'Company B';
```

```
run;
```

